#
#
#

isEmpty(QMKBLD_PRJCFG_INCLUDED) { # include guard
QMKBLD_PRJCFG_INCLUDED = 1

#
# Main project build modes
#
PRJ_CONFIG = debug
#PRJ_CONFIG = release

# Project wide coverage support
#PRJ_COVERAGE = 1

# Project wide profiling support
#PRJ_PROFILE = 1

#	 Are we in release mode?
release {
  # Enable inline stament in TC code base
  DEFINES += TC_USE_INLINE
}

#
# Sun Studio specific items
#
contains(QMK_COMP,Sunpro) {

# Printing compiler steps
#PRJ_CONFIG *= ss_compstep

# Stack overflow check
#PRJ_CONFIG *= ss_stkovf

# Coverage support
#PRJ_CONFIG *= ss_tcov

# Thread analyzer support
#PRJ_CONFIG *= ss_tha

!isEmpty(PRJ_COVERAGE): PRJ_CONFIG *= ss_tcov

}

#
# Windows specific configuration
#
win32 {
 PRJ_CONFIG += console
 # Are we in debug mode?
 debug {
   # Use iterator debugging
   DEFINES *= _SECURE_SCL=1
   DEFINES *= _SECURE_SCL_THROWS=1
   # Use Run-time checks for stack corruption and uninitialized var use
   QMAKE_CXXFLAGS += /RTC1
 }
} # end win32
else {
 !contains(TEMPLATE,subdirs) {
   MOC_DIR = moc
   OBJECTS_DIR = obj
	 QMAKE_CLEAN += moc/* obj/*
   dclean.commands = rm -fr moc obj
   dclean.depends = distclean
	 QMAKE_EXTRA_TARGETS += dclean
 }
}




isQmkDebug() {
	message(PRJ_CONFIG = $$PRJ_CONFIG)
  win32: QMAKE_LFLAGS += /verbose:lib
}

# Now transfer project configuration tokens into CONFIG
CONFIG += $$PRJ_CONFIG

# This project depends on qt-cvt project
CVT_INC = $$(PRJS_HOME)/qt-cvt/inc
CVT_LIB = $$(PRJS_HOME)/qt-cvt/lib

# In the .pro file followings should be specified:
#INCLUDEPATH += $$CVT_INC
#LIBS += -L$$CVT_LIB -lgtm -lcvt
#


# Base library should be on the path
DEPENDPATH *= $$PRJ_SRC/base
INCLUDEPATH *= $$PRJ_SRC/base


} # end of include guard